Arduino workshop A37 kort verslag 7, 5 april 2019.

Op het programma staat: Rotary Encoder.
Het is vanavond bijzonder rustig maar evengoed is er goed overlegd, uitgewisseld en afgekeken. 

Piet heeft al eerder op zijn website een Rotary Encoder library gezet, daarin zitten 4 voorbeelden (zie verder).
Ook heeft Piet kort geleden een sketch (WorkshopClock1.ino) voor de klok met potmeter / drukknop    en een 2-tal aansluitschema’s op zijn website gezet.
Zoals ondertussen gebruikelijk, een aantal uitwerkingen van mij en van Henk, PA0HPV als bijlage.
Bij het verlaten van de ruimte memoreerde Piet nog dat volgende keer het de laatste keer is!
//// WorkshopClock1
// install RTC

#include <SparkFunDS1307RTC.h>

#include <wire.h>

//install LCD

#include <LiquidCrystal.h>

const int rs = 12, en = 11, d4 = 7, d5 = 6, d6 = 5, d7 = 4;

LiquidCrystal lcd(rs, en, d4, d5, d6, d7); //pins used by LCD

#define ROWS 2

#define COLUMS 16

#define pushed !notPushed

void printADay(int day);

void printAMonth(int month);

bool show;

int day, month, cursor; 

int filler, notPushed;

int pot, toSet, toSetCursor, cursorSet, stateSet, toVal;

unsigned int seconds, minutes, hours, days, dates, months, years;

unsigned int elapsedTime, currentTime;

void setup() {

  lcd.begin(COLUMS, ROWS);

  lcd.clear();

  rtc.begin();

  Serial.begin(9600);

  pinMode(A2,INPUT_PULLUP); //pushbutton pin

  notPushed=digitalRead(A2);

  stateSet=0;

  currentTime=millis();

  elapsedTime=0;

}//end setup

void loop() 

{

  currentTime=millis();

  if(currentTime<elapsedTime)elapsedTime=0;

  if((currentTime-elapsedTime)>200)show=true;

  if(show)

  {

    show=false;     

    elapsedTime=currentTime; 

    if(stateSet==0)lcd.noBlink();

    rtc.update();

    seconds = rtc.second();

    minutes = rtc.minute();

    hours = rtc.hour();

    days = rtc.day();

    dates = rtc.date();

    months = rtc.month();

    years = rtc.year();

    lcd.setCursor(1,0);

    if(stateSet==4 && cursorSet==0)printADay(toVal);else printADay(days);

    lcd.print(" ");

    if(stateSet==4 && cursorSet==1 )dates=toVal;

    if(dates<10)lcd.print(" ");

    lcd.print(dates);

    lcd.print(" "); 

    if(stateSet==4 && cursorSet==2)printAMonth(toVal);else printAMonth(months);

    lcd.print(" 20");

    //if(years<10)lcd.print(0); //only required for 2100 and beyond

    if(stateSet==4 && cursorSet==3)lcd.print(toVal); else lcd.print(years);

    lcd.setCursor(4,1);

    if(stateSet==4 && cursorSet==4)hours=toVal;

    if(hours<10)lcd.print(0);

    lcd.print(hours);

    lcd.print(":");

    if(stateSet==4 && cursorSet==5)minutes=toVal;

    if(minutes<10)lcd.print(0);

    lcd.print(minutes);

    lcd.print(":");

    if(stateSet==4 && cursorSet==6)seconds=toVal;

    if(seconds<10)lcd.print(0);

    lcd.print(seconds);

    notPushed=digitalRead(A2);

    switch(stateSet)

    {

      case 0: lcd.noBlink();

              if(pushed)stateSet=1; break;

      case 1: if(notPushed) stateSet=2; break;

      case 2: if(notPushed)  

              {

                lcd.blink();

                toSetCursor = (map(analogRead(A1), 0, 1023, 0, 6));

                switch(toSetCursor)

                {

                  case 0: lcd.setCursor(1,0);break;

                  case 1: lcd.setCursor(4,0);break;

                  case 2: lcd.setCursor(7,0);break;

                  case 3: lcd.setCursor(13,0);break;

                  case 4: lcd.setCursor(4,1);break;

                  case 5: lcd.setCursor(7,1);break;

                  case 6: lcd.setCursor(10,1);break;

                  default: ;

                }

              cursorSet=toSetCursor;  

              }

              else stateSet=3;  

              break;

      case 3: if(notPushed) stateSet=4; break;

      case 4: if(notPushed)   

              {  

                switch(cursorSet) 

                {

                  case 0: toVal=map(analogRead(A1), 0, 1023, 1, 8); 

                          if(toVal>7)toVal=7; 

                          break; //day

                  case 1: toVal=map(analogRead(A1), 0, 1023, 1, 32); if(toVal>31)toVal=31; break; //date

                  case 2: toVal=map(analogRead(A1), 0, 1023, 1, 13); if(toVal>12)toVal=12; break; //month

                  case 3: toVal=map(analogRead(A1), 0, 1023, 1, 100); if(toVal>99)toVal=99; break; //year

                  case 4: toVal= map(analogRead(A1), 0, 1023, 0, 24); if(toVal>=23)toVal=23; break; //hour

                  case 5: toVal= map(analogRead(A1), 0, 1023, 0, 60); if(toVal>59)toVal=59; break; //minute

                  case 6: toVal= map(analogRead(A1), 0, 1023, 0, 60); if(toVal>59)toVal=59; break; //second     

                }

              }

              else stateSet=5;

              break;

      case 5: if(notPushed) stateSet=6; break;       

      case 6: if(notPushed)

              {

                switch(cursorSet)

                {

                  case 0: rtc.setDay(toVal); break;

                  case 1: rtc.setDate(toVal); break;

                  case 2: rtc.setMonth(toVal); break;

                  case 3: rtc.setYear(toVal); break;

                  case 4: rtc.setHour(toVal); break;

                  case 5: rtc.setMinute(toVal); break;

                  case 6: rtc.setSecond(toVal); break;       

                }                

              }

              else stateSet=0;

              break;

     } // end switchSet   

  } //end show 

}//end loop

void printADay(int day)

{

  switch(day)

  {

    case 1: lcd.print("zo");break;

    case 2: lcd.print("ma");break;

    case 3: lcd.print("di");break;

    case 4: lcd.print("wo");break;

    case 5: lcd.print("do");break;

    case 6: lcd.print("vr");break;

    case 7: lcd.print("za");break;

    default:lcd.print("??");break;

  }

}// end printADay

void printAMonth(int month)

{

   switch(month)

   {

     case 1: lcd.print("jan");break;

     case 2: lcd.print("feb");break;

     case 3: lcd.print("mrt");break;

     case 4: lcd.print("apr");break;

     case 5: lcd.print("mei");break;

     case 6: lcd.print("jun");break;

     case 7: lcd.print("jul");break;

     case 8: lcd.print("aug");break;

     case 9: lcd.print("sep");break;

     case 10: lcd.print("okt");break;

     case 11: lcd.print("nov");break;

     case 12: lcd.print("dec");break;

     default: lcd.print("???");break;   

   }

}// end printAMonth
RotaryEncoder   4 voorbeeld sketches

// -----

// InterruptRotator.ino - Example for the RotaryEncoder library.

// This class is implemented for use with the Arduino environment.

// Copyright (c) by Matthias Hertel, http://www.mathertel.de

// This work is licensed under a BSD style license. See http://www.mathertel.de/License.aspx

// More information on: http://www.mathertel.de/Arduino

// 18.01.2014 created by Matthias Hertel

// -----

// This example checks the state of the rotary encoder in the loop() function.

// The current position is printed on output when changed.

// Hardware setup:

// Attach a rotary encoder with output pins to A2 and A3.

// The common contact should be attached to ground.

#include <RotaryEncoder.h>

// Setup a RoraryEncoder for pins A2 and A3:

RotaryEncoder encoder(A2, A3);

void setup()

{

  Serial.begin(57600);

  Serial.println("SimplePollRotator example for the RotaryEncoder library.");

  // You may have to modify the next 2 lines if using other pins than A2 and A3

  PCICR |= (1 << PCIE1);    // This enables Pin Change Interrupt 1 that covers the Analog input pins or Port C.

  PCMSK1 |= (1 << PCINT10) | (1 << PCINT11);  // This enables the interrupt for pin 2 and 3 of Port C.

} // setup()

// The Interrupt Service Routine for Pin Change Interrupt 1

// This routine will only be called on any signal change on A2 and A3: exactly where we need to check.

ISR(PCINT1_vect) {

  encoder.tick(); // just call tick() to check the state.

}

void loop()   // Read the current position of the encoder and print out when changed.

{

  static int pos = 0;

  int newPos = encoder.getPosition();

  if (pos != newPos) {

    Serial.print(newPos);

    Serial.println();

    pos = newPos;

  // Just to show, that long lasting procedures don't break the rotary encoder:

  // When newPos is 66 the output will freeze, but the turned positions will be recognized even when not polled.

  // The interrupt still works.

  // The output is correct 6.6 seconds later.

  if (newPos == 66)

    delay(6600);

  } // if

} // loop ()

// The End

// -----

// LimitedRotator.ino - Example for the RotaryEncoder library.

// This class is implemented for use with the Arduino environment.

// Copyright (c) by Matthias Hertel, http://www.mathertel.de

// This work is licensed under a BSD style license. See http://www.mathertel.de/License.aspx

// More information on: http://www.mathertel.de/Arduino

// 26.03.2017 created by Matthias Hertel

// -----

// This example checks the state of the rotary encoder in the loop() function.

// The current position is printed on output when changed.

// In addition to the SimplePollRotator example here the range of the rotator is limited to the range 0 - 16 and only incremental steps of 2 are realized.

// To implement this limit the boundaries are checked and eventually the current position is adjusted.

// The internal (physical) position of the rotary encoder library remains by stepping with the increment 1

// so the the logical position is calculated by applying the ROTARYSTEPS factor.

// Hardware setup:

// Attach a rotary encoder with output pins to A2 and A3.

// The common contact should be attached to ground.

#include <RotaryEncoder.h>

#define ROTARYSTEPS 2

#define ROTARYMIN 0

#define ROTARYMAX 16

RotaryEncoder encoder(A2, A3);   // Setup a RoraryEncoder for pins A2 and A3:

int lastPos = -1;   // Last known rotary position.

void setup()

{

  Serial.begin(57600);

  Serial.println("LimitedRotator example for the RotaryEncoder library.");

  encoder.setPosition(10 / ROTARYSTEPS); 
// start with the value of 10.

} // setup()

void loop()   // Read the current position of the encoder and print out when changed.

{

  encoder.tick();

  // get the current physical position and calc the logical position

  int newPos = encoder.getPosition() * ROTARYSTEPS;

  if (newPos < ROTARYMIN) {

    encoder.setPosition(ROTARYMIN / ROTARYSTEPS);

    newPos = ROTARYMIN;

  } else if (newPos > ROTARYMAX) {

    encoder.setPosition(ROTARYMAX / ROTARYSTEPS);

    newPos = ROTARYMAX;

  } // if

  if (lastPos != newPos) {

    Serial.print(newPos);

    Serial.println();

    lastPos = newPos;

  } // if

} // loop ()

// The End

// -----

// SimplePollRotator.ino - Example for the RotaryEncoder library.

// This class is implemented for use with the Arduino environment.

// Copyright (c) by Matthias Hertel, http://www.mathertel.de

// This work is licensed under a BSD style license. See http://www.mathertel.de/License.aspx

// More information on: http://www.mathertel.de/Arduino

// -----

// 18.01.2014 created by Matthias Hertel

// -----

// This example checks the state of the rotary encoder in the loop() function.

// The current position is printed on output when changed.

// Hardware setup:

// Attach a rotary encoder with output pins to A2 and A3.

// The common contact should be attached to ground.

#include <RotaryEncoder.h>

// Setup a RoraryEncoder for pins A2 and A3:

RotaryEncoder encoder(A2, A3);

void setup()

{

  Serial.begin(57600);

  Serial.println("SimplePollRotator example for the RotaryEncoder library.");

} // setup()

// Read the current position of the encoder and print out when changed.

void loop()

{

  static int pos = 0;

  encoder.tick();

  int newPos = encoder.getPosition();

  if (pos != newPos) {

    Serial.print(newPos);

    Serial.println();

    pos = newPos;

  } // if

} // loop ()

// The End

// -----

// SimplePollRotatorLCD.ino - Example for the RotaryEncoder library.

// This class is implemented for use with the Arduino environment.

// Copyright (c) by Matthias Hertel, http://www.mathertel.de

// This work is licensed under a BSD style license. See http://www.mathertel.de/License.aspx

// More information on: http://www.mathertel.de/Arduino

// -----

// 18.01.2014 created by Matthias Hertel

// -----

// This example checks the state of the rotary encoder in the loop() function.

// The current position is printed on output when changed.

// Hardware setup:

// Attach a rotary encoder with output pins to A2 and A3.

// The common contact should be attached to ground.

#include <Wire.h> 

#include <LiquidCrystal_PCF8574.h>

#include <RotaryEncoder.h>

// Setup a RoraryEncoder for pins A2 and A3:

RotaryEncoder encoder(A2, A3);

LiquidCrystal_PCF8574 lcd(0x27);  
// set the LCD address to 0x27 for a 16 chars and 2 line display

void setup()

{

  lcd.begin(16, 2); 


// initialize the lcd 

  lcd.setBacklight(1);

  lcd.print("SimplePollRotatorLCD");

} // setup()

// Read the current position of the encoder and print out when changed.

void loop()

{

  static int pos = 0;

  encoder.tick();

  int newPos = encoder.getPosition();

  if (pos != newPos) {

    lcd.setCursor(0, 1); 

    lcd.print(newPos);

    lcd.print(" ");

    pos = newPos;

  } // if

} // loop ()

// The End

RotaryEncoder.h
// -----

// RotaryEncoder.h - Library for using rotary encoders.

// This class is implemented for use with the Arduino environment.

// Copyright (c) by Matthias Hertel, http://www.mathertel.de

// This work is licensed under a BSD style license. See http://www.mathertel.de/License.aspx

// More information on: http://www.mathertel.de/Arduino

// -----

// 18.01.2014 created by Matthias Hertel

// -----

#ifndef RotaryEncoder_h

#define RotaryEncoder_h

#include "Arduino.h"

#define LATCHSTATE 3

class RotaryEncoder

{

public:

  // ----- Constructor -----

  RotaryEncoder(int pin1, int pin2);

  // retrieve the current position

  long  getPosition();

  // simple retrieve of the direction the knob was rotated at. 0 = No rotation, 1 = Clockwise, -1 = Counter Clockwise

  int8_t getDirection();

  // adjust the current position

  void setPosition(long newPosition);

  // call this function every some milliseconds or by using an interrupt for handling state changes of the rotary encoder.

  void tick(void);

private:

  int _pin1, _pin2; 


// Arduino pins used for the encoder. 

  volatile int8_t _oldState;

  volatile long _position;         

// Internal position (4 times _positionExt)

  volatile long _positionExt;      

// External position

  volatile long _positionExtPrev;  
// External position (used only for direction checking)

};

#endif

// End

Mijn combinatie van voorbeeld 2 en 4, aangevuld met een aantal zaken (LCD uitlezing / LED’s aanduiding)

/// RotaryEncoder_Aad1

// -----

//// combinatie van 2 voorbeelden uit de library. aangevuld met tekst en een aantal LED’s bij bepaalde standen
//// LimitedRotator.ino & SimplePollRotatorLCD.ino - Example for the RotaryEncoder library.

// This class is implemented for use with the Arduino environment.

// Copyright (c) by Matthias Hertel, http://www.mathertel.de

// This work is licensed under a BSD style license. See http://www.mathertel.de/License.aspx

// More information on: http://www.mathertel.de/Arduino

// 26.03.2017 created by Matthias Hertel

// Ontwikkeld op 24 maart 2019, door Aad, PE1HTJ i.v.m. een Arduino-workshop bij Veron A37

// -----

// This example checks the state of the rotary encoder in the loop() function.

// The current position is printed on output when changed.

// The range of the rotator is limited and incremental steps are realized.

// To implement this limit the boundaries are checked and eventually the current position is adjusted.

// The internal (physical) position of the rotary encoder library remains by stepping with the increment 1

// so the the logical position is calculated by applying the ROTARYSTEPS factor.

// Hardware setup:

// Attach a rotary encoder with output pins to A2 and A3.

// The common contact should be attached to ground.

#include <RotaryEncoder.h>

#include <Wire.h> 

#include <LiquidCrystal.h>

#define ROTARYSTEPS 1    

////  2

#define ROTARYMIN 0      

////  0

#define ROTARYMAX 20      

////  16

RotaryEncoder encoder(A2, A3);   

// Setup a RoraryEncoder for pins A2 and A3:

int lastPos = -1;                


// Last known rotary position.

/// LiquidCrystal  LCD1602

const int rs = 12, en = 11, d4 = 5, d5 = 4, d6 = 3, d7 = 2;

LiquidCrystal lcd(rs, en, d4, d5, d6, d7);

int ledR = 7;  



// led red    op pin 7

int ledY = 8;  



// led yellow op pin 8

int ledG = 9;  



// led green  op pin 9

//////////////////////////////  set up  ////////////////////////////////////////

void setup()

{

pinMode(ledR, OUTPUT);   

pinMode(ledY, OUTPUT);  

pinMode(ledG, OUTPUT);  

  Serial.begin(57600);

  Serial.println("Simpele Rotaty Encoder Met LCD door Aad, PE1HTJ.");

  encoder.setPosition(10 / ROTARYSTEPS); 

// start with the value of 10.

  lcd.begin(16, 2);




// initialize the lcd 

  lcd.print("RotaryEncoderLCD");  


////lcd.print("SimpelRotEnc LCD");

  lcd.setCursor(0, 1); 

  lcd.println("door Aad, PE1HTJ.");

  delay(5000);

} // eind setup()

//////////////////////////////  loop  ///////////////////////////////////////////

// Read the current position of the encoder and print out when changed.

void loop()

{ 

  encoder.tick();

  // get the current physical position and calc the logical position

  int newPos = encoder.getPosition() * ROTARYSTEPS;

  if (newPos < ROTARYMIN) {

    encoder.setPosition(ROTARYMIN / ROTARYSTEPS);

    newPos = ROTARYMIN;

  } else if (newPos > ROTARYMAX) {

    encoder.setPosition(ROTARYMAX / ROTARYSTEPS);

    newPos = ROTARYMAX;

  } // eind   else if

///////////////////// UITLEZINGEN  //////////////////

//

////////////////// op seriele monitor  //////////////

  if (lastPos != newPos) {

    Serial.println("* *");       



/// scheiding tussen regels

    Serial.print(lastPos); Serial.println(" is de vorige stand");

    Serial.print(newPos);  Serial.print(" is nu de huidige stand");

      if (newPos==0) { Serial.print(" en minimaal ! !"); }

      if (newPos==1) { Serial.print(", bijna minimaal !"); }

      if (newPos==10) { Serial.print(": de begin stand !"); }

      if (newPos==19) { Serial.print(", bijna maximaal !"); }

      if (newPos==20) { Serial.print(" en maximaal ! !"); }

    Serial.println();         /// naar nieuwe regel

/////////////// op display  en  LED's  /////////////////////

    lcd.clear();    

    lcd.setCursor(0, 0); lcd.print("  oud:");

    lcd.setCursor(6, 0); lcd.print(lastPos); 

    lcd.setCursor(0, 1); lcd.print("nieuw:");

    lcd.setCursor(6, 1); lcd.print(newPos);

    lcd.setCursor(8, 1); 

      if (newPos==0) { lcd.print(" - min.!"); 

        digitalWrite(ledG, HIGH); delay(2000); 

        digitalWrite(ledG, LOW); }

      if (newPos==1) { lcd.print(" <- min."); 

        digitalWrite(ledY, HIGH); delay(300); 

        digitalWrite(ledG, HIGH); delay(1000); 

        digitalWrite(ledG, LOW); digitalWrite(ledY, LOW); }

      if (newPos==10) { lcd.print(" begin");  

        digitalWrite(ledY, HIGH); delay(500); 

        digitalWrite(ledY, LOW); delay(500); 

        digitalWrite(ledY, HIGH); delay(500); 

        digitalWrite(ledY, LOW); }

      if (newPos==19) { lcd.print(" -> max."); 

        digitalWrite(ledY, HIGH); delay(300); 

        digitalWrite(ledR, HIGH); delay(1000); 

        digitalWrite(ledR, LOW); digitalWrite(ledY, LOW); }

      if (newPos==20) { lcd.print(" + max.!"); 

        digitalWrite(ledR, HIGH); delay(2000); 

        digitalWrite(ledR, LOW); }

//////////////////////////

    } // eind   if (lastPos

    lastPos = newPos;       

/// nieuwe waarde overnemen en opnieuw uitlezen

} // eind loop ()

// The End
Bijdrage van  PA0HPV, Henk

/* Arduino Rotary Encoder op LCD

 * by Dejan Nedelkovski, www.HowToMechatronics.com

 * van internet gehaald en aangepast 

 * feb/mrt 2019 door PE1HTJ, Aad en PA0HPV, Henk.

 * verbind encoder output A met pin 6 v.d. nano

 * verbind encoder output B met pin 7 v.d. nano

 * de drukknopfunctie van de rotary encoder wordt nog niet gebruikt

 */

 #include "LiquidCrystal.h"

 LiquidCrystal lcd(12, 11, 5, 4, 3, 2);

#define outputA 6         // CLK

#define outputB 7         // DT

// #define ButtonPin 8    // SW

int counter = 0;

int aState = 0;

int aLastState = 0;

void setup() {

  pinMode (outputA, INPUT);

  pinMode (outputB, INPUT);

//  pinMode (ButtonPin, INPUT_PULLUP);

  lcd.begin(16,2);

  lcd.setCursor(4,0);

  lcd.print("Positie:");

  lcd.setCursor(4,1);

  lcd.print(counter); // anders begint het prg zonder cijfer

  aLastState = digitalRead(outputA);

} // eind void setup()

void loop() {

  aState = digitalRead(outputA);

  if (aState != aLastState){

    if (digitalRead(outputB) != aState){

      counter ++;

    } else {

      counter --;

    }

    if(counter<0){

    lcd.setCursor(3,1); // minteken komt een plaats naar links te staan

    } else {

    lcd.setCursor(3,1);

    lcd.print(" ");

    lcd.setCursor(4,1);}

    lcd.print(counter);

    delay(20);

    lcd.print("      ");

    }

    aLastState = aState;

//  hier kan je dus verder gaan op grond van de waarde van  aState

//  if (aState = waarde)   ///enz.

//  ook de drukknop kan gebruikt worden na declaratie

//  if (digitalRead(ButtonPin) == "LOW")   /// enz.

} // eind void loop()

11

